
CHAPTER 8
Analytics and Output Reporting

S
ophisticated financial models are virtually useless if the results are inaccurate or
difficult to understand. The best financial models have internal tests that track

computational and conceptual accuracy. While the purpose of these tests is to make
sure that the model calculation is valid, the interpretation of the calculations needs
to be created and presented in a comprehensive and simple-to-understand format.
The goal is to have a single output sheet that summarizes the model calculations and
provides metrics for financial performance.

INTERNAL TESTING

Internal tests are the first step in reporting results. They make sure that as a model is
modified and changed, the core logic still functions as originally intended. The tests
also allow a user to troubleshoot problems faster if the need arises. Tests mainly
focus on cash flow calculations, but allow a user to interpret conceptual validity
based on the test results. This will become clearer as internal tests are implemented
in Project Model Builder.

A number of tests will be set up in Project Model Builder to make sure that the
model is functioning correctly. Each one is discussed as it is implemented.

Cash In versus Cash Out

One of the most fundamental tests is to make sure that whatever cash has gone
into the deal has come out. This can be reworded in a more important manner: to
make sure that whatever cash is used in the deal was funded by cash coming into
the transaction. Essentially, a cash flow model has a finite amount of money from
the assets. The liabilities can only draw from this finite amount.

A critical error that sometimes occurs is when a model is modified and cells are
not linked correctly cash is created or used from a nonexistent source. To make
sure this does not occur the cash that comes into a deal should be tracked and
compared against the cash that flows out of the deal. Except for extraordinarily rare
circumstances there should never be a difference.

MODEL BUILDER 8.1: CASH IN VERSUS CASH OUT TEST

1. Most tests will be tracked on the Cash Flow sheet, but reported on the Inputs
sheet and eventually the Output sheet when it is created. For cash in versus cash
out, columns will need to be created on the Cash Flow sheet.
Go to column CJ of the Cash Flow sheet and enter these labels in the following
cells:
CJ4: Cash In
CK4: Cash Out
CL4: Difference

2. The Cash In is all of the cash that is available to pay liabilities. At first this is
all of the cash that the assets generate. While that is a large part of the Cash
In each period, two of the advanced features of the structure provide cash: the
swap and the reserve account. In cell CJ7 enter:

= Q7 + R7 + T7 + U7 + AG7 + BI7

Tracing each one of these back, the cash flow coming into the transaction
consists of voluntary prepayments, scheduled asset amortization, asset yield,
recoveries from defaulted assets, swap flows in, and any amount coming in from
the reserve. Copy this formula and paste it over the range CJ7:CJ366.

3. The Cash Out has more references since there many points that cash comes out
of the transaction. Enter the following formula in cell CK7:

= AI7+AN7+AT7+AV7+BA7+BC7+BJ7+BP7+BU7+BY7+BZ7

The formula is self-explanatory for simple liabilities such as fees paid (cell
AN7), but notice some of the less obvious references such as cells AV7, BC7,
BJ7, BY7, and BZ7. Swap payments sometimes go out so these must be deducted.
Remember, too, that reserve account withdrawals were considered to be Cash
In, so the actual use of that cash to cover liabilities is Cash Out. Also, if the
reserve is reimbursed cash leaves the transaction. Finally, if there is excess cash
at the end of the waterfall it is used by either applying the cash to senior principal
or releasing it. Copy and paste cell CK7 over the range CK7:CK366.

4. The real test now is to see if the Cash In minus the Cash Out is equal to zero.
Enter the following formula in cell CL7:

= CJ7−CK7

Copy and paste this formula over the range CL7:CL366. See Figure 8.1 for the
new section on the Cash Flow sheet.

5. Checking the entire ‘‘Difference’’ column each run would be tedious, so some
quick links need to be set up. First sum up range CL7:CL366 in cell CL5 by using:

= SUM(CL7:CL366)

FIGURE 8.1 Cash In versus Cash Out helps
prevent funds from being ‘‘created’’ in the
model.

6. Next go to the Inputs sheet. Since most of the model is controlled from this
sheet it is useful to make sure the model is running correctly as assumptions are
changed. Label cell I3 TESTS and I4 Cash In = Cash Out. Go to cell L4 and
enter the following formula:

= IF('Cash Flow'!CL5=0,"OK","ERROR")

This formula checks the sum of the cash in versus cash out differences to see if
it is zero. If it is then the model is working correctly, otherwise there is an error.
Many of the tests will be set up using this IF statement format with either an OK
or ERROR return. Conditional formatting is particularly useful here so that it is
very obvious when there is an error. If unfamiliar with conditional formatting,
see the ToolBox section of this chapter. Otherwise make the cell shading green
and font bold white for when the cell is OK and red shaded with bold white
font when there is an ERROR.

FIGURE 8.2 The TESTS section is on the Inputs sheet for quick reference.

Balances at Maturity

Two very important tests check the asset and liability balance at maturity. The more
important of the two is the debt balance at maturity. If a principal amount remains
unpaid at maturity, most likely the debt balance will incur a loss. These tests are
very quick to implement.

MODEL BUILDER 8.2: BALANCES AT MATURITY TESTS

1. Go to the Inputs sheet and enter the following labels:
I5: Asset Balance @ Maturity
I6: Senior Debt @ Maturity

2. In cell L5 enter the following formula:

= IF('Cash Flow'!V366<1,"OK","ERROR")

This checks the final possible period for the assets. If there is a balance greater
than $1, than there is a problem. Use the same conditional formatting on cell
L5 as seen in the first test for this chapter.

3. The senior debt balance will be looked at with more scrutiny since it is the focus
of further analysis. To make automating the model easier in Chapter 10, go to
the Cash Flow sheet and name cell CB366: FinalLoanBal.

4. In cell L6 enter the following formula:

= IF(FinalLoanBal<1,"OK","ERROR")

If the senior debt is not paid off by the final period, then an error will show. Use
the same conditional formatting as seen in the first test for this chapter. Note
that this test can be done for the subordinate debt if needed, but is only done
for the senior debt in Project Model Builder. See Figure 8.3 for detail.

FIGURE 8.3 The Balance tests are integral to the model and are readily seen on the Inputs
sheet.

Asset Principal Check

In a system where there is a finite amount of cash coming from the assets, it is
imperative that the cash inflow is correct. As asset amortization goes beyond simple
amortization and exotic products with unusual loss curves or payment methods
are incorporated, it is important to make sure the individual components of asset
amortization equal the original amount. In other words, does the sum of all the
prepayments, scheduled amortization, and defaulted amounts equal the beginning
balance of the assets?

Remember that prepayments, scheduled amortization, and defaults reduce the
asset balance each month. If this is the case, then summing up all of those reductions
should equal the beginning balance. An error often occurs here when there are data
problems with the representative lines or in a loan level model, with the individual
loans. Also a loss or prepay curve could be off or a custom payment might be
calculated incorrectly.

MODEL BUILDER 8.3: ASSET PRINCIPAL CHECK TEST

1. Go to the Cash Flow sheet and make sure that there are sums of the voluntary
prepay, actual amort, and new default columns. This is done by the following
formulas:

N5: = SUM(N7:N366)

Q5: = SUM(Q7:Q366)

R5: = SUM(R7:R366)

2. Go back to the Inputs sheet and enter the following label in cell I7: Asset
Principal Check. In cell L7 enter the following formula:

= IF(ROUND(('Cash Flow'!N5+'Cash Flow'!Q5+'Cash Flow'!R5),0)

= ROUND('Cash Flow'!V6,0),"OK","ERROR")

This formula adds up all the sums of the asset reduction and compares that sum
to the original asset balance. The ROUND function is used because as assets are
calculated there can be very minor differences in the decimal positions that will
falsely trip this test. Also use the same conditional formatting on cell L7 as seen
in the first test of this chapter.

FIGURE 8.4 The Tests section of the Inputs sheet is complete.

PERFORMANCE ANALYTICS

The Cash Flow sheet is impractical to quickly garner information from unless
additional calculations are performed. The calculations should explain relevant cha-
racteristics for financial analysis. All parties to a transaction are primarily concerned
with yield, loss, and the timing of cash flows. These concepts are captured by monthly
yield, bond equivalent yield, duration, and weighted average life calculations.

Monthly Yield

Monthly yield is more importantly used to calculate bond equivalent yield, rather
than as a metric on its own. The monthly yield of an asset or debt is the discount
rate that makes the present value of all of the cash flow from the asset or to the debt
equivalent to the initial principal balance. For assets, the cash flow that is counted
is the yield, scheduled amortization, voluntary prepayments, and the recovered
principal. For debt, the cash flow that is counted is the interest and principal.

MODEL BUILDER 8.4: CALCULATING MONTHLY YIELD

1. Many of the performance analytics require a stream of cash flows for the assets
and debt that is discounted. Instead of doing this on the cash flow sheet, create
a new sheet named Analytics.

2. Since most of the model is complete, there are many sections that can be
referenced instead of recreating formulas. In cell A13 enter:

= 'Cash Flow'!A4

Drag cell A13 over the range A13:C375. This will reference the dates and timing
section from the Cash Flow sheet. The range A14:C15 on the Analytics sheet
will have zero value references, so clear the contents of these cells.

3. To the right of the dates and timing will be the discounted cash flows for the
assets and the debt. Label the following cells by entering these cell references:
E13: = AssetDes1
F13: = LiabDes1
G13: = LiabDes2
Column D will be left blank for a space between the dates and timing, and the
discounted cash flows.

4. Before calculating the discounted cash flows, a few more items need to be set
up. Enter the following references to populate labels:
E3: = AssetDes1
F3: = LiabDes1
G3: = LiabDes2

5. In cells B4 and B5, enter the following labels respectively: Initial Principal and
Monthly Yield.

6. The initial principal amounts are needed for the calculations and should be
referenced as follows:
E4: = AssetCurBal1
F4: = 'Cash Flow'!CB6
G4: = 'Cash Flow'!CF6

7. For now enter a starting monthly yield of 1.0% in cells E5, F5, and G5. Also,
for automation purposes later name the range E5:G5, rngYieldChange.

8. Next the discounted cash flows need to be calculated. For the assets enter the
following formula in cell E16:

= ('Cash Flow'!Q7+'Cash Flow'!R7+'Cash Flow'!T7

+'Cash Flow'!U7)/(1 + $E$5)∧$A16

This formula adds the voluntary prepayments, amortization, interest, and recov-
eries as cash flow from the assets. That sum is then discounted by the assumed
monthly yield. Copy this formula and paste it over the range E16:E375.

9. The discounted cash flow for the debt is similar, but the cash flows associated
with them are interest and principal. Enter the following formulas in cells F16
and G16:

F16: = ('Cash Flow'!CC7 + 'Cash Flow'!CD7)/(1 + $F$5)∧$A16

G16: = ('Cash Flow'!CG7 + 'Cash Flow'!CH7)/(1 + $G$5)∧$A16

Copy and paste these formulas over their respective ranges (F16:F375 and
G16:G375).

10. Go back up to cell B6 and enter the label PV Difference. The present value
difference is the result of subtracting the sum of the present valued cash flow

from the initial balance. For the assets and debt enter the following formulas:

E6: = SUM(E16:E375)-E4

F6: = SUM(F16:F375)-F4

G6: = SUM(G16:G375)-G4

Name the range E6:G6, rngYieldTarget.

Calculating the Monthly Yield

The yield calculations are in place, but to really understand what is happening
look at the PV difference. For the assets and debt, there should be a negative PV
difference. This means that the discount rate, which is the assumed monthly yield, is
too high. Try reducing the assumed monthly yield to .50 percent for each. Now there
is a positive PV difference, which means that the assumed monthly yield is too low.

Trying higher and lower values is an inefficient means of what is technically
known as a divide-and-conquer algorithm. Excel has a built-in tool called Goal
Seek that can perform the operation to find the exact monthly yield necessary. If
Goal Seek is unfamiliar, see the Toolbox section of this chapter. Otherwise, try goal
seeking the asset monthly yield as an example.

Open the Goal Seek tool and make the ‘‘Set Cell’’ referenced to the PV difference
in cell E6. The ‘‘To value’’ should be set to zero because the correct monthly yield
will make the present value of the cash flows equal to the initial balance. Subtracting
the same amount from one another should produce zero. Finally, the ‘‘By changing
cell’’ is the monthly yield because that is the value that needs to iterate until the PV
difference is zero. Once these parameters are entered as seen in Figure 8.5, run Goal

FIGURE 8.5 Excel’s Goal Seek tool is used to find the monthly yield.

Seek. The monthly yield should change until the PV difference is zero. Note that
there could be errors in trying to calculate the yield of an asset or liability that is
taking a complete loss and that the yield should just be assumed to be zero.

While this goal seek process should be repeated for both debt tranches, it is very
tedious and has to be redone every time a change to the cash flow is made. Later in
the text, VBA automation is used to control this process and allows a user to click a
single button to calculate the yield for the assets and all debt tranches.

Bond-Equivalent Yield

A standard convention in the capital markets is to report yield in terms of the
bond-equivalent yield (BEY), which can be thought of as an annual yield. This is
achieved by multiplying the semiannual yield by two. Debate exists about the best
annual yield measure, but since convention uses BEY, it should be incorporated into
the model.

MODEL BUILDER 8.5: CALCULATING BOND-EQUIVALENT YIELD

1. On the Analytics sheet, in cell B7 enter the label BEY. To calculate the BEY,
the semiannual yield needs to be calculated from the monthly yield. Do this by
entering the following formula in cell E7:

= ((1+E5)∧6 − 1)

2. As stated earlier, the BEY is just the semiannual yield doubled. Modify the
formula so it reads:

=2*((1+E5)∧6 − 1)

3. Copy and paste this formula over the range E7:G7. The asset and debt BEYs
should be very similar to the average annual interest rates of the assets and debt
tranches.

MODIFIED DURATION

Duration measures a bond value’s sensitivity to rate changes. Fabozzi officially
defines it as ‘‘the approximate percentage change in value for a 100 basis point
change in rates.’’1 The formula for modified duration is:

1

(1 + yield/k)

[

1∗PVCF1 + 2∗PVCF2 + · · · + n∗PCVCFn

k∗Price

]

1Frank J. Fabozzi, Fixed Income Analysis. (New Hope, PA: Frank J. Fabozzi Associates,
2000), 255.

where

k = periodicity of the payments

n = total periods

PVCFt = present value of the cash flow in period t discounted by the yield

MODEL BUILDER 8.6: CALCULATING MODIFIED DURATION

1. On the Analytics Sheet, label cell B8: Modified Duration. The formula for
modified duration translates easily into Excel. In cell E8 enter:

= 1/((1+E5/12))*(SUMPRODUCT(E16:E375,$A$16:$A$375)/(12*E4))

2. Most of the formula is self-explanatory with the written equation presented
earlier. The one item to notice is that the SUMPRODUCT function is used on
the present values of the cash flows for more efficient calculation. Copy and
paste this formula over the range E7:G7. The final Analytics section should look
like Figure 8.6.

FIGURE 8.6 The analytics sheet is now complete.

OUTPUT REPORTING

Most people do not have the time to sift through the minutiae of a complex financial
model. Users of financial models generally want an understanding of results pertinent
to their decision-making process. A majority of these results can be summarized in
one page.

This single page, known as an output report, most efficiently conveys a cash
flow model when it is split into three separate sections: scenario assumptions, cash
flow snapshot, and charts. Figure 8.7 provides a general overview of an output sheet.
These three sections make it easy for a person to quickly see what assumptions were
put into the model, what the results are for important periods in the cash flow, and
how certain data intensive assumptions/results look graphically in charts.

An important part of the output report is that it will be created as its own
worksheet in the model and should contain as few entered values as possible.
This means that nearly all of the sheet should reference parts of the model. The
reasoning behind this is that as assumptions and results change per scenario, the
output report should automatically update. Errors usually occur when a scenario is
run, but the output sheet is not updated. By using as many references as possible
there will be fewer possibilities for an error and an increase in model operation
efficiency.

FIGURE 8.7 The output sheet has three distinct sections: Scenario Assumptions, Cash Flow
snapshot, and Charts.

MODEL BUILDER 8.7: CREATING THE OUTPUT REPORT

1. The first step to creating the Output Report is to make a section on the Inputs
sheet for reporting parameters. On the Inputs sheet, enter these labels in the
following cells:
G28: Reporting Parameters
G29: Scenario Name
G30: Beginning Snapshot Start
G31: Ending Snapshot Start

2. For now, enter Base Case in cell I29 and name it ScenName. Enter 0 in cell
I30 and name it SnapshotStart. Finally, enter 350 in cell I31 and name it
SnapshotEnd. These cells are used later to control certain parts of the output
report.

3. Create a new worksheet and name it Output. Since this sheet is primarily
referenced to other parts of the model, the step by step will work differently
than the other Model Builder sections. Rather than explain each cell, when
most of them are references to concepts and calculations already explained, this
section will rely heavily on the already completed version of file MB8-7.xls in
the Ch08 folder on the CD-ROM.

4. In the completed version of Model Builder 8.7, notice that the first nine rows
consist primarily of assumptions. A reader instantly knows which transaction
the report has been created for by looking at the top left in bold letters. At the
top right, the scenario version is readily visible. Below that the first few boxes
a reader sees are inputs relating to the assets, liabilities, and structure. Almost
all of these are references that should be linked up individually for the model
under construction. Create all of the references for the first nine rows exactly
the same as in the completed version. There are a few cells with formulas that
require further explanation.

5. A unique cell on the Output sheet is I5, the gross cumulative loss. Instead of
a cell reference to the default rate, this is a calculation of the gross cumulative
dollar defaults by the original asset balance. In this case the gross cumulative
loss percent is .97 percent, while the default rate is 1.0 percent. The difference
can be occurring due to seasoning, amortization timing, and/or loss timing.
Understanding this nuance will be discussed further in Chapter 9, Understanding
the Model. For now it is important to realize that the gross cumulative loss
percent should not be a direct reference.

6. Another unique cell is O4, which is labeled Req. Cr Enhmnt. In a senior
subordinated structure with one senior tranche, this is the amount of credit
enhancement that is necessary to absorb the expected losses. For now enter:

= 1−F4

This simply subtracts the senior advance rate from 100 percent and suggests
that anything below the senior advance rate is the amount necessary to cover

the expected losses. This is precisely how the senior tranche should be sized.
The next chapter will discuss sizing the senior tranche in more detail.

7. Cells O5 and O6 contain formulas for weighted average life (‘‘WAL’’), a metric
that is used frequently. This calculation was left off of the Analytics sheet
because it does not rely on discounted cash flows. Instead WAL is a calculation
based on the amortization of the assets and liabilities over time. It expresses the
number of periods it takes to amortize an asset or liability, weighted by periodic
reductions of balance.

Imagine an asset that has equal periodic reductions in balance of 150 and
pays off in 10 periods. In this case a total of 1,500 has been paid down equally
over 10 periods. The WAL would be 5.5 periods, which is just the same as
taking the average of the number of periods. However, if the assets paid down
with 600 in the first period, followed by 300 in the second, 200 in the third,
100 in the fourth, and 50 for the remaining six periods, the WAL would be
2.97 periods. In such a case, the balance was reduced by a large amount earlier
and has ‘‘died off’’ faster. See Figure 8.8 and the WALifeComparison.xls file in
Ch08’s Additional Files subfolder on the CD-ROM for details.

Prior to inputting the formulas in cells O4 and O5, a column tracking all of
the reductions in asset balance is necessary on the Cash Flow sheet. Go to the
Cash Flow sheet to column CN. Enter Asset Amort Total in cell CN4. Then in
cell CN7 enter:

= N7 + Q7 + R7

FIGURE 8.8 The WAL is a measure of how long an assets balance is
outstanding.

This formula adds up any amount that reduces the asset balance: defaults,
scheduled principal, and voluntary prepayments. Copy and paste this formula
over the range CN7:CN366.

8. Calculating WAL is a straightforward weighted average formula that uses the
SUMPRODUCT-SUM combination seen in previous chapters. Back on the
Output sheet, in cell O5, enter:

= SUMPRODUCT('Cash Flow'!CN7:CN366,'Cash Flow'!A7:A366)/

SUM('Cash Flow'!CN7:CN366)/12

This weights the periods by the asset reduction each period and then provides
an average. Also notice that there is a 12 as a divisor since the WAL is typically
presented in years and this model is being calculated on a monthly basis.

9. The WAL should also be calculated for the debt. On this output sheet only the
senior debt has the WAL calculated. Since the only item that reduces debt balance
is principal, an additional column on the Cash Flow sheet is not necessary. In
cell O6 enter:

= SUMPRODUCT('Cash Flow'!CD7:CD366,'Cash Flow'!A7:A366)/

SUM('Cash Flow'!CD7:CD366)/12

10. The last unique formula for the upper portion of the Output sheet tracks when
any trigger has been tripped in the model. To accomplish this functionality, a
single column on the Cash Flow sheet that tracks whether or not any trigger has
been tripped in any period needs to be created.
Go to the Cash Flow sheet to column CO. Label cell CO4 Combined Trigger
Tracking. Then in cell CO7 enter:

= OR(Z7:AB7)

This will return a TRUE or FALSE depending on whether any of the triggers
in Z, AA, or AB have been tripped. Copy and paste cell CO7 over the range
CO7:CO366.

11. Once the periodic tracking has been set up, a formula needs to return the first
period that the trigger tracking becomes TRUE. This can be done using an array
formula. If array formulas are new, see the Toolbox section at the end of this
chapter. The array formula used in cell O9 should be:

{= MIN(IF('Cash Flow'!CO7:CO366,'Cash Flow'!A7:A366,1000))}

The inside part of this formula is an IF statement that checks each cell in the
range CO7:CO366. If any of those are true the associated period in column A
is returned, otherwise a very large value that exceeds the maximum number of

periods is returned (1,000). With a string of periods where the trigger is tripped
the earliest one can be found using the MIN function. Remember to press Ctrl
+ Shift + Enter when entering an array function.

12. The middle part of the output report provides a snapshot of the cash flows.
Reams of complete cash flows are rarely looked through, but specific periods
such as the beginning and end are often examined. To do this a system needs to
be set up that allows a user to select the periods of the cash flows to view.
Earlier, a Reporting Parameters section was added to the Inputs sheet, where
cells I30 and I31 contain controls to change the first period for the beginning
and ending views of the cash flows. Go to the Outputs sheet to cell B16. This
will be a direct reference to cell I30 on the Inputs sheet (named SnapshotStart).
On the Outputs sheet, cell B17 is the next period, which is the previous period
plus one. Enter the following formula in cell B17:

= B16+1

Make sure not to use the named range because when cell B17 is copied and
pasted over the range B16:B26 the reference needs to change. By now B17:B26
should look like Figure 8.9.

13. Still on the Outputs sheet, go to C16. The completed version shows the beginning
balance for this column. It should be noted that any of the cash flow columns
can be shown on the Outputs sheet and that the ones used in the completed
version are just being used because they are the most frequently used. To get the

FIGURE 8.9 The Cash
Flow snapshot relies on a
select group of periods
determined by the model
operator.

beginning balances of the assets for the periods used in the snapshot viewer, use
an OFFSET function. Enter the following formula in cell C16:

= OFFSET('Cash Flow'!$V$6,B16,0)

This formula offsets the cell above the first period of the asset balance on the
Cash Flow sheet by the period used in the snapshot viewer (cell B16). As a user
changes the starting period for the snapshot beginning, the asset balance will be
offset commensurately. Copy and paste cell C16 over the range C16:C26.
This can be repeated for as many columns as needed. Look to the completed
Output sheet to see some other popular columns that are reported. Also notice
that there are sums for each column that are references to the Cash Flow sheet
directly. This is because the sums should be for all of the periods and should not
just sum the snapshot view.

14. Complete the same process for the ending snapshot view in rows 28 to 38 on the
Output sheet. Leave row 27 blank so it is clear where the beginning snapshot
ends and where the ending snapshot begins.

15. The final section of the output report is the bottom third, which should
contain charts of data over time. Examples of such data include: interest rates,
prepayment rates, default rates, balances, excess spread, and so on. Some of the
charts may not have source data readily available and will require more work
on other sheets. For example, in the completed version of the output report
there is a chart that tracks cumulative defaults. The source data for this chart is
from column CP on the Cash Flow sheet.

The Importance of Testing and Output

While it may seem tedious to implement all of the tests and the output report, the
time saved in the long run is much greater. Whenever building a model keep the
mindset that it will be used for multiple scenarios. Each time the model assumptions
are changed an operator wants to be sure that the results are accurate, logical,
and easy to share. Tests ensure accuracy and logic, and should be viewed easily.
Similarly, the output report should be comprehensive enough so someone who has
not seen the model can understand the intent, execution, and results of each scenario
the model produces.

TOOLBOX

Conditional Formatting

Conditional formatting automatically changes the formatting of cells depending on
parameters that a user sets up. This can be especially useful to quickly identify
important changes that occur during model operation.



Analytics and Output Reporting 141

To use conditional formatting, select the desired cell, click Format on the menu
bar and then click Conditional Formatting. The Conditional Formatting dialog box
appears as shown in Figure 8.10.

The Conditional Formatting dialog box allows for a number of parameters.
First, the format can depend on the value of the cell. This is set up by selecting
Cell Value Is in the first Condition 1 list box and then a series of corresponding
parameter options in the other list boxes. The other option is to select Formula Is in
the first Condition 1 list box and select a formula. Multiple formats can also be set
up for a single cell. This is particularly useful in conjunction with IF statements to
highlight a change in true or false values.

Goal Seek

Excel provides the Goal Seek tool that allows a user to perform single parameter
optimization operations. Goal Seek works by taking in three assumptions: a formula
that needs to be optimized, the value that the result of the formula should be
optimized to, and a variable that changes which allows the original formula to be
optimized.

The Goal Seek dialog box shown in Figure 8.11 is opened by clicking Tools in
the menu bar and then Goal Seek.

FIGURE 8.10 Conditional Formatting dialog box.

FIGURE 8.11 Goal Seek dialog box.

FIGURE 8.12 Enter the cell references in the Goal Seek dialog box.

As an example, if a sheet had 10 in cell A1, 2 in cell A2, and the formula A1*A2
in A3, Goal Seek can be used to figure out an optimal solution. With the current
setup, cell A3 is equal to 20. If one wanted cell A3 to equal 30 by changing the
value in cell A2, Goal Seek could be used. In that case, the references and values in
Figure 8.12 would be entered in the dialog box. Also refer to GoalSeekExample.xls
in Ch08’s Additional Files subfolder on the CD-ROM.

As will be seen later in this text, Goal Seek can be used on an entire model
with many formulas connected to each other. Also, Excel contains a more advanced
optimization tool called Solver when multiple parameters need to be entered.

Array Formulas

Array formulas are one of the more tricky aspects for Excel users to learn. The main
reason people have difficulty with them is because they reference ranges of cells,
which can be challenging to visualize. An array function allows a user to perform
custom operations on arrays of equal lengths.

Deconstructing the array function used in this chapter is an excellent method to
demonstrate how they work. In this chapter the following array function is used:

{= MIN(IF('Cash Flow'!CO7:CO366,'Cash Flow'!A7:A366,1000))}

The array function here first performs an IF statement on the range CO7:CO366.
The values for column CO are true or false values. In a regular IF statement, a single
cell value is tested to return one of two answers. In the array function version, every
cell in the array is tested against a parameter or range of parameters and returns an
equal length result array. If cell CO7 is false, for example, then 1,000 is returned
as the first value of the result array. If cell CO8 is true, then A8 is returned as the
second value of the result array. However, if cell CO9 is false, then cell A9 will be
returned as the third value of the result array. The first three values of the result
array can be visualized as 1,000, 1, 2. The next part of this array function takes the
MIN of that array, which is 1.

There are two important rules to remember when using an array function. Any
arrays referenced within the formula must be of equal length; otherwise an error
will be returned. Also, array functions must be entered by pressing Ctrl + Shift
+ Enter instead of simply Enter. An array function that is successfully entered can
be identified with the { } symbols surrounding it.


